COMPASS for Pressure
Test Macro to log 2456-LEM Laboratory Environment Monitor outputs to a file
[bookmark: _GoBack]Use with this file that has a header row in it, if not used a file without header row will be created
C:\dhi\COMPASS for Pressure\Data\2456-LEM logs\LEM log file.log

Copy and Paste all of the following text to a Test Macro in the COMPASS Macro Editor

'Test Macros do not have a return value.  Manipulate
'the test or device collection as desired.
'iT  The current temperature point in the test
'iL  The current line pressure point in the test.
'iC  The current pressure cycle in the test.
'IP  The current pressure point in the test.
'cTest   The test class .
'cConfig  Configuration class that holds all active devices.
'Author: Kyle Clark, 22-Jun-2016

Function LogAmbToFile(iT, iL, iC, iP, cTest, cConfig)
On Error Resume Next
Dim sText, LogAmbFile, logDate, PrsAmb, TempAmb, RH_Amb

'This is a macro that logs data to an outside datafile that is semicolon delimited
'The outside file name will be created if it does not exist
LogAmbFile = "C:\dhi\COMPASS for Pressure\Data\2456-LEM logs\LEM log file.log"

logDate = True    '"True" will log date and time on each line, "False" will not

PrsAmb = cCompass.cConfig.AmbPrs.FinalOutput & ";" & cCompass.cConfig.AmbPrs.RangeMain.UnitFinalText & ";"
'	cDebug.LogStatus "AmbPrs = " & PrsAmb
TempAmb = cCompass.cConfig.AmbTemp.FinalOutput & ";" & cCompass.cConfig.AmbTemp.RangeMain.UnitFinalText & ";"
'	cDebug.LogStatus "AmbTemp " & TempAmb
RH_Amb = cCompass.cConfig.AmbRH.FinalOutput & ";" & cCompass.cConfig.AmbRH.RangeMain.UnitFinalText
'	cDebug.LogStatus "AmbRH " & RH_Amb
sText = PrsAmb & TempAmb & RH_Amb

'Use Function (macro) below to write to file
LogLEMtoTextFile LogAmbFile, sText, logDate, ";"

End Function

'//////////////////////////////////////////////
'Open/Create a text file for Append and add the 
'passed-in text to the file.  Optionally date/time 
'stamp each entry, based on the boolean bLogDate 
'parameter. Separator defines how the fields are 
'separated, a ; is used by default.
'Close the file after each write, so it is available
'from multiple processes.

Function LogLEMtoTextFile(fName, sText, bLogDate, Separator) 
Dim fso, f, sDate

Set fso = CreateObject("Scripting.FileSystemObject")
Set f = fso.OpenTextFile(fName, 8, True)	'8 = open for append

If Separator = "" Then Separator = ";"		'default separator to semi-colon

If bLogDate Then
	sDate = Date & Separator & formatDateTime(Time,3) & Separator	'3 = vbLongTime
	f.Write vbCrLf & sDate & sText
Else
	f.Write vbCrLf & sText
End If
f.Close

LogLEMtoTextFile = True 

End Function
